Имеется: 1С:Предприятие 8.2 (8.2.15.301) УПП и Самописная(Делопроизводство).
Из УПП будем получать информацию о Контрагентах, и добавлять в Самописную. Все действия происходят в Самописной.
Цель: достучаться до Адреса и телефона Контрагентов в УПП.
Подключаюсь к другой базе через V82.COMConnector.
Перебором могу получить информацию из справочника Контрагенты (включая Адреса и телефоны).
А вот запросом не получается. Вернее запрос отрабатывает, можно получить наименования и т.п. информацию. Но вот Адреса и телефон выводятся как пустые поля.

```1c
    Запрос.Текст = " ВЫБРАТЬ
    | Контрагенты.Ссылка Как Ссылка,
    | Контрагенты.ИНН,
    | Контрагенты.КПП,
    | Контрагенты.Наименование,
    | Контрагенты.НаименованиеПолное,
    | КИ_Телефон.Представление КАК Телефон,
    | КИ_Адрес.Представление КАК АдресЮридический,
    | КИ_Адрес2.Представление КАК АдресФактический,
    | Контрагенты.ОсновнойВидДеятельности.Наименование КАК ВидДеятельности,
    | Контрагенты.ПометкаУдаления,
    | Контрагенты.Родитель
    |ИЗ
    | Справочник.Контрагенты КАК Контрагенты
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Телефон
    | ПО (КИ_Телефон.Объект = Контрагенты.Ссылка)
    | И (КИ_Телефон.Тип = &Т1)
    | И (КИ_Телефон.Вид = &В1)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес
    | ПО (КИ_Адрес.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес.Тип = &Т2)
    | И (КИ_Адрес.Вид = &В2)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес2
    | ПО (КИ_Адрес2.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес2.Тип = &Т3)
    | И (КИ_Адрес2.Вид = &В3)
    |ГДЕ
    | НЕ Контрагенты.ПометкаУдаления
    | И НЕ Контрагенты.ЭтоГруппа";

    Запрос.УстановитьПараметр("Т1","Телефон");
    Запрос.УстановитьПараметр("В1","Телефон контрагента");
    Запрос.УстановитьПараметр("Т2","Адрес");
    Запрос.УстановитьПараметр("В2","Юридический адрес контрагента");
    Запрос.УстановитьПараметр("Т3","Адрес");
    Запрос.УстановитьПараметр("В3","Фактический адрес контрагента");
```

Данный запрос сам по себе рабочий. В УПП получаем Всю информацию.
Когда выполняем этот же запрос, но через СОМ, поля Телефон, АдресЮридический и АдресФактический возвращаются пустыми. На сколько смог разобраться, всё упирается в "КИ_Телефон.Объект = Контрагенты.Ссылка" - а вернее Контрагенты.Ссылка. Ссылки в Самописной базе ни что иное, как COM-объект... Подскажите пожалуйста, как победить сиё безобразие?

PS. Простите за маленькое вознаграждение, всё что имеется :(
---
Выкладываю конечный вариант. Себе, что бы не забыть, другим как помощь!

```1c
    СOM1с=Новый COMObject("V82.COMConnector");
    Попытка
    ВнешняяБаза = СOM1с.Connect("Srvr=serv1cv82;Ref=workbase;Usr=Пользователь;Pwd=Пароль123");
    //serv1cv82 - Кластер серверов 1С:Предприятия,
    //workbase - Имя информационной базы в кластере
    //Пользователь и Пароль - для доступа к 1С.
    //Сообщить("База подключена");
    Исключение
    Предупреждение("Ошибка открытия базы!!!");
    Сообщить(ОписаниеОшибки());
    ВнешняяБаза=Неопределено;
    Возврат;
    КонецПопытки;

    Запрос = ВнешняяБаза.NewObject("Запрос");
    Запрос.Текст = " ВЫБРАТЬ
    | ПРЕДСТАВЛЕНИЕ(Контрагенты.Ссылка) Как Ссылка,
    | Контрагенты.ИНН,
    | Контрагенты.КПП,
    | Контрагенты.Наименование,
    | Контрагенты.НаименованиеПолное,
    | КИ_Телефон.Представление КАК Телефон,
    | КИ_Адрес.Представление КАК АдресЮридический,
    | КИ_Адрес2.Представление КАК АдресФактический,
    | Контрагенты.ОсновнойВидДеятельности.Наименование КАК ВидДеятельности,
    | Контрагенты.ПометкаУдаления,
    | Контрагенты.Родитель
    |ИЗ
    | Справочник.Контрагенты КАК Контрагенты
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Телефон
    | ПО (КИ_Телефон.Объект = Контрагенты.Ссылка)
    | И (КИ_Телефон.Тип = &Т1)
    | И (КИ_Телефон.Вид = &В1)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес
    | ПО (КИ_Адрес.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес.Тип = &Т2)
    | И (КИ_Адрес.Вид = &В2)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес2
    | ПО (КИ_Адрес2.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес2.Тип = &Т3)
    | И (КИ_Адрес2.Вид = &В3)
    |ГДЕ
    | НЕ Контрагенты.ПометкаУдаления
    | И НЕ Контрагенты.ЭтоГруппа";

    Запрос.УстановитьПараметр("Т1",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Телефон);
    Запрос.УстановитьПараметр("В1",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ТелефонКонтрагента);
    Запрос.УстановитьПараметр("Т2",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Адрес);
    Запрос.УстановитьПараметр("В2",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ЮрАдресКонтрагента);
    Запрос.УстановитьПараметр("Т3",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Адрес);
    Запрос.УстановитьПараметр("В3",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ФактАдресКонтрагента);

    Результат = Запрос.Выполнить();
    Выборка = Результат.Выбрать();

    Счетчик = 0;
    Пока Выборка.Следующий() Цикл
    Счетчик = Счетчик + 1;
    Сообщить("Ссылка: "+СокрЛП(Выборка.Ссылка)+" | ИНН: "+СокрЛП(Выборка.ИНН)+" | КПП: "+СокрЛП(Выборка.КПП)+" | Наименование: "+ СокрЛП(Выборка.Наименование)+" | Полное Наименование: "+ СокрЛП(Выборка.НаименованиеПолное)+"
    | Телефон: "+СокрЛП(Выборка.Телефон)+" | АдресЮридический: "+СокрЛП(Выборка.АдресЮридический)+" | АдресФактический: "+СокрЛП(Выборка.АдресФактический)+"
    | ");
    Если Счетчик >= 50 Тогда
    Прервать;
    КонецЕсли;
    КонецЦикла;

    ВнешняяБаза = Неопределено;
```
---
KrackMan, а точно в регистре КонтактнаяИнформация поля Вид и Тип имеют строковый тип?

Ведь ты в качестве параметров для этимх полей передаешь строку (Запрос.УстановитьПараметр("Т1","Телефон");), что страно.

У тебя этот запрос в исходной базе если его там запустить (с этими же параметрами) нужную информацию возвращает?
---
Ягг, "У тебя этот запрос в исходной базе если его там запустить нужную информацию возвращает?"
Да, нужную. Но с одной оговоркой, все делается в конструкторе запроса.

"Ведь ты в качестве параметров для этимх полей передаешь строку (Запрос.УстановитьПараметр("Т1","Телефон");), что страно. "
Сейчас сделаю обработку в УПП, о результате отпишусь.
Спасибо за подсказку :)
---
KrackMan, обрати внимаение на типы параметров в конструкторе запроса, там скорее всего ссылки на соотвествующие справочники (перечисления), а не строковые переменные.
---
Ягг, (7) andrewks, вы оба правы. Тестовая обработка в УПП показала, что дело действительно в текстовых параметрах. Так же выводит поля пустыми. А я уже неделю не в том направлении копаю ...

PS. Как вознаграждение отдать?
---
KrackMan, не смартманькой единой жив человек :)
тебе и самому пригоднятся еще :)
---
Ягг, Тогда человеческое спасибо! :)
---
используйте соединение таблиц в запросе
---
andrewks, а с учетом оконченного текста?
---
KrackMan, думаю, нужно юзать не строки, а Справочники.ВидыКонтактнойИнформации
---
Выкладываю конечный вариант. Себе, что бы не забыть, другим как помощь!

```1c
    СOM1с=Новый COMObject("V82.COMConnector");
    Попытка
    ВнешняяБаза = СOM1с.Connect("Srvr=serv1cv82;Ref=workbase;Usr=Пользователь;Pwd=Пароль123");
    //serv1cv82 - Кластер серверов 1С:Предприятия,
    //workbase - Имя информационной базы в кластере
    //Пользователь и Пароль - для доступа к 1С.
    //Сообщить("База подключена");
    Исключение
    Предупреждение("Ошибка открытия базы!!!");
    Сообщить(ОписаниеОшибки());
    ВнешняяБаза=Неопределено;
    Возврат;
    КонецПопытки;

    Запрос = ВнешняяБаза.NewObject("Запрос");
    Запрос.Текст = " ВЫБРАТЬ
    | ПРЕДСТАВЛЕНИЕ(Контрагенты.Ссылка) Как Ссылка,
    | Контрагенты.ИНН,
    | Контрагенты.КПП,
    | Контрагенты.Наименование,
    | Контрагенты.НаименованиеПолное,
    | КИ_Телефон.Представление КАК Телефон,
    | КИ_Адрес.Представление КАК АдресЮридический,
    | КИ_Адрес2.Представление КАК АдресФактический,
    | Контрагенты.ОсновнойВидДеятельности.Наименование КАК ВидДеятельности,
    | Контрагенты.ПометкаУдаления,
    | Контрагенты.Родитель
    |ИЗ
    | Справочник.Контрагенты КАК Контрагенты
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Телефон
    | ПО (КИ_Телефон.Объект = Контрагенты.Ссылка)
    | И (КИ_Телефон.Тип = &Т1)
    | И (КИ_Телефон.Вид = &В1)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес
    | ПО (КИ_Адрес.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес.Тип = &Т2)
    | И (КИ_Адрес.Вид = &В2)
    | ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.КонтактнаяИнформация КАК КИ_Адрес2
    | ПО (КИ_Адрес2.Объект = Контрагенты.Ссылка)
    | И (КИ_Адрес2.Тип = &Т3)
    | И (КИ_Адрес2.Вид = &В3)
    |ГДЕ
    | НЕ Контрагенты.ПометкаУдаления
    | И НЕ Контрагенты.ЭтоГруппа";

    Запрос.УстановитьПараметр("Т1",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Телефон);
    Запрос.УстановитьПараметр("В1",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ТелефонКонтрагента);
    Запрос.УстановитьПараметр("Т2",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Адрес);
    Запрос.УстановитьПараметр("В2",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ЮрАдресКонтрагента);
    Запрос.УстановитьПараметр("Т3",ВнешняяБаза.Перечисления.ТипыКонтактнойИнформации.Адрес);
    Запрос.УстановитьПараметр("В3",ВнешняяБаза.Справочники.ВидыКонтактнойИнформации.ФактАдресКонтрагента);

    Результат = Запрос.Выполнить();
    Выборка = Результат.Выбрать();

    Счетчик = 0;
    Пока Выборка.Следующий() Цикл
    Счетчик = Счетчик + 1;
    Сообщить("Ссылка: "+СокрЛП(Выборка.Ссылка)+" | ИНН: "+СокрЛП(Выборка.ИНН)+" | КПП: "+СокрЛП(Выборка.КПП)+" | Наименование: "+ СокрЛП(Выборка.Наименование)+" | Полное Наименование: "+ СокрЛП(Выборка.НаименованиеПолное)+"
    | Телефон: "+СокрЛП(Выборка.Телефон)+" | АдресЮридический: "+СокрЛП(Выборка.АдресЮридический)+" | АдресФактический: "+СокрЛП(Выборка.АдресФактический)+"
    | ");
    Если Счетчик >= 50 Тогда
    Прервать;
    КонецЕсли;
    КонецЦикла;

    ВнешняяБаза = Неопределено;
```
---
Очень помогли !